Universelle FileAuswahldialoge für Ihre Datenbanken, Teil 1
Versionen: Access 2003, 2002/XP, 2000 und 97

Zum Helpline-Beitrag "PCX-Bilder in Datenbank laden" in Access Weekly vom 17.2.2006 haben uns zahlreiche Zuschriften erreicht, in denen es um die Verwendung eines flexiblen Dateiauswahl-Dialogs anstelle der festen Zuweisung eines Dateinamens ging. Wir möchten Ihnen daher in einem zweiteiligen Tipp der Woche fertige Lösungen für die drei wichtigsten immer wieder benötigten Auswahldialoge vorstellen.

Beginnen wir mit einigen Grundlagen: Während Sie in Word oder Excel die dort vorhandenen Dialoge zum Öffnen oder Speichern von Dateien in Ihren VBA-Lösungen einsetzen können, ist das in Access nicht bzw. erst ab Access 2002/XP mit dem Objekt "FileDialogs" möglich. "FileDialogs" erlaubt zwar auch die Anzeige eines Dialogs zur Auswahl eines Ordners, aber dazu wird ein modifizierter Dateiauswahl-Dialog verwendet, der nicht dem unter Windows gewohnten "Ordner öffnen"-Dialog entspricht. Wir beginnen deshalb mit einem dem Windows-Pendant entsprechenden "Ordner öffnen"-Dialog, den Sie in allen Access-Versionen einsetzen können, sowie dem "FileDialogs"-Gegenstück, das nur in Access 2003 und 2002/XP genutzt werden kann. Im zweiten Teil folgen dann die Datei-Dialoge "OpenFile()" und "SaveFile()" in Varianten für Access 2003 und 2002/XP bzw. Access 2000 und 97.

Ab Windows 95 stehen dem Entwickler verschiedene API-Funktionen zur Vereinheitlichung der Benutzeroberfläche und der Benutzerführung zur Verfügung. So soll sichergestellt werden, dass beispielsweise Dateien oder Verzeichnisse über immer gleiche Auswahl-Dialoge gesucht oder Farben, Schriften und Drucker über immer gleiche Optionen-Dialoge eingestellt werden können, ohne dass der Anwender sich für jede Anwendung in einen anderen Dialog einarbeiten muss. Für die Auswahl von Verzeichnissen ist hier die Funktion "SHBrowseForFolder()" zuständig. Der Zusatz "SH" macht eine so genannte "Shell-Funktion", also eine Benutzeroberflächen-Funktion, kenntlich. "SHBrowseForFolder" lässt sich leider nicht direkt aufrufen, da die Funktion mit einer speziellen Datenstruktur arbeitet, die vor dem Aufruf initialisiert werden muss. Das erfolgt in der Funktion "OpenFolder()", die Sie zunächst samt den notwendigen Deklarationen in ein Modul "modOpenFolder" kopieren:

rivate Type BROWSEINFO
    hwndOwner As Long
    pidlRoot As Long
    lpszDisplayName As String
    lpszTitle As String
    ulFlags As Long
    BFFCALLBACK As Long
    LPARAM As Long
    iImage As Integer
End Type

Private Declare Function SHBrowseForFolder Lib _
        "shell32.dll" (FolderStruct As BROWSEINFO) As Long

Private Declare Function SHGetPathFromIDList Lib _
        "shell32.dll" (ByVal LPCITEMIDLIST As Long, _
        ByVal lpStr As String) As Long

Function OpenFolder(strMsg) As String
  Dim BI As BROWSEINFO
  Dim R As Long
  Dim lpBuffer As String * 254

  With BI
    .hwndOwner = 0
    .lpszDisplayName = lpBuffer
    .lpszTitle = strMsg
    .ulFlags = 0
    .BFFCALLBACK = 0
    .LPARAM = 0
  End With
  R = SHBrowseForFolder(BI)
  R = SHGetPathFromIDList(R, lpBuffer)
  OpenFolder = Left$(lpBuffer, InStr(lpBuffer, _
  Chr(0)) - 1)

End Function

Mit der Zuweisung ".lpszDisplayName = lpBuffer" setzen wir zunächst den Puffer für die Rückmeldung des gewählten Verzeichnisnamens. Nach dem Aufruf wird auf diesem Weg nur der Name des gewählten Verzeichnisses, nicht aber dessen kompletter Pfad geliefert! Wenn also beispielsweise "Z:\Test\Daten\Access" ausgewählt wird, enthält dieses Feld anschließend nur "Access". Wir werten diese Rückmeldung nicht weiter aus, sondern geben "lpBuffer" quasi nur als "Dummy-Variable" an. Anschließend setzen wir mit ".lpszTitle = strMsg" den Text, der als Parameter "strMsg" übergeben wurde und der im Dialog als Hinweis für den Anwender angezeigt werden soll. Der Aufruf per "R = SHBrowseForFolder(BI)" sorgt dann für die Anzeige des Dialoges, wobei die eben initialisierte Datenstruktur als Parameter übergeben wird. Beim Funktionsergebnis in "R" handelt sich um eine spezielle Struktur, die über eine Zusatzfunktion "SHGetPathFromIDList()" ausgewertet werden muss.

Diese Struktur ist zwar als "String" deklariert, es handelt sich aber um eine kompliziert aufgebaute "ItemList", für die es in VBA keinen entsprechenden Datentyp gibt. Die Funktion "SHGetPathFromIDList()" extrahiert aus der über "R" referenzierten Liste nun den reinen Pfad des ausgewählten Verzeichnisses und legt diesen in "lpBuffer" ab. Beim Ergebnis handelt es sich um einen "C"-String, also eine mit "0" terminierte Zeichenkette, so dass eine letzte Anweisung "OpenFolder = Left$(lpBuffer, InStr(lpBuffer, Chr(0)) - 1)" alle Zeichen links vom "0"-Byte der Zeichenkette extrahiert. Das Ergebnis ist dann wie bereits erwähnt, eine leere Zeichenkette nach einem Klick auf Abbrechen oder der Pfad als VBA-String.

In Ihren VBA-Routinen können Sie die Funktion "OpenFolder()" dann zum Beispiel wie folgt nutzen:

strVerzeichnis = _
  OpenFolder("Bitte Export-Verzeichnis wählen:")
If strVerzeichnis = "" Then
  Exit Sub 'Nichst gewählt...
Else
  'Hier Ihre Verarbeitung...
End If

Das Ergebnis der Funktion "OpenFolder()" ist entweder eine leere Zeichenkette "", wenn auf Abbrechen geklickt wurde oder der komplette Pfad des ausgewählten Verzeichnisses wie zum Beispiel "Z:\Test\Datenaustausch\Excel". Das Ergebnis wird immer ohne abschließenden Backslash geliefert!

Mit Office 2002 hat Microsoft wie bereits erwähnt das Objekt "FileDialogs" eingeführt, das ohne externe API-Funktionen und komplexe Datenstrukturen auskommt und zudem komfortabler zu handhaben ist. Es sei aber nochmals angemerkt, dass Sie die oben gezeigte Funktion "OpenFolder()" auch problemlos mit Access 2003 und 2002/XP einsetzen können! Die auf "FileDialogs" basierende Funktion "OpenFolder()" ist wie folgt aufgebaut:

Function OpenFolder(strMsg As String, _
                    strPath As String) As String
  Dim fDlg As FileDialog

  Set fDlg = _
      Application.FileDialog(msoFileDialogFolderPicker)
  With fDlg
    .ButtonName = "Export"
    .Title = strMsg
    If Right$(strPath, 1) <> "\" Then _
      strPath = strPath + "\"
    End If
    .InitialFileName = strPath
    .Show
    If .SelectedItems.Count > 0 Then
      OpenFolder = .SelectedItems(1)
    Else
      OpenFolder = ""
    End If
  End With

End Function

Zunächst muss eine Objektvariable "fDlg" vom Typ "FileDialog" über "Set fDlg=..." initialisiert werden, um den weiteren Zugriff darüber zu ermöglichen. Über die Eigenschaften "Title" und "ButtonName" können dann individuelle Bezeichnungen für den Dialog und die Ok-Schaltfläche zugeordnet werden. Über "InitialFileName" können Sie ein bestimmtes Verzeichnis vorgeben. Der Aufruf der Methode "Show" zeigt den Dialog dann an. Das Ergebnis ist eine Auflistung "SelectedItems", über deren Eigenschaft "Count" sich zunächst abfragen lässt, ob etwas ausgewählt oder ob auf Abbrechen (Count=0) geklickt wurde. Das Auslesen des ersten Eintrages der Auflistung liefert dann den kompletten Pfad des gewählten Verzeichnisses bzw. wenn "Count = 0" zutrifft, eine leere Zeichenkette als Ergebnis.

In Ihren VBA-Routinen können Sie die Funktion dann zum Beispiel wie folgt nutzen:

strVerzeichnis = _
  OpenFolder("Bitte Export-Verzeichnis wählen:", _
             "Z:\Test\Daten")
If strVerzeichnis = "" Then
  Exit Sub 'Nichts zu tun...
Else
  'Hier Ihre Verarbeitung...
End If

Im ersten Teil dieses Tipps der Woche hatten wir Ihnen zunächst Lösungen für die Anzeige einer Ordner-Auswahl basierend auf den Windows Common Controls und dem Access 2002/2003-Objekt "FileDialogs" vorgestellt. Im zweiten Teil geht es nun um Funktionen zum Thema "Datei öffnen" und "Datei speichern". Solche Funktionen benötigen Sie regelmäßig, um beispielsweise die Anwender einer Datenbank eine zu importierende Datei wählen oder um das Ziel für einen Export festlegen zu lassen. Auch hier sind wieder verschiedene Lösungen auf Basis der Windows Common Controls und dem Access 2002/2003-Objekt "FileDialogs" möglich.

Schauen wir zunächst einmal, wie Funktionen wie "OpenFile()" und "SaveFile()" mit Hilfe der Windows Common Controls umgesetzt werden können. Hier stehen die API-Funktionen "GetOpenFilename()" und "GetSaveFilename()" zur Verfügung. Beide Funktionen arbeiten mit einer Datenstruktur "OPENFILENAME", über die vor dem Aufruf verschiedene Eigenschaften wie das Startverzeichnis für die Auswahl, der Text für die Titelleiste oder die Einträge für das Dropdown-Menü "Dateityp" spezifiziert werden. Nach dem Aufruf enthält diese Struktur den gewählten Dateinamen inklusive Laufwerk und Pfad. Um die Initialisierung und Auswertung der Datenstruktur zu vereinfachen, erfolgt der Aufruf über zwei Funktionen "OpenFile()" und "SaveFile", denen als Parameter jeweils das Startverzeichnis für die Auswahl sowie der Text für die Titelleiste übergeben wird.

Um diese Funktionen in Ihren Datenbanken zu nutzen, legen Sie ein neues Modul "modDatDlgs" oder ähnlich an und kopieren die folgenden Deklarationen und Funktionen in das Modul:

Type OPENFILENAME
  lStructSize As Long
  hwndOwner As Long
  hInstance As Long
  lpstrFilter As String
  lpstrCustomFilter As String
  nMaxCustFilter As Long
  nFilterIndex As Long
  lpstrFile As String
  nMaxFile As Long
  lpstrFileTitle As String
  nMaxFileTitle As Long
  lpstrInitialDir As String
  lpstrTitle As String
  flags As Long
  nFileOffset As Integer
  nFileExtension As Integer
  lpstrDefExt As String
  lCustData As Long
  lpfnHook As Long
  lpTemplateName As String
End Type

Private Const OFN_FILEMUSTEXIST = &H1000
Private Const OFN_HIDEREADONLY = &H4
Private Const OFN_OVERWRITEPROMPT = &H2
Private Const OFN_PATHMUSTEXIST = &H800

Private Declare Function GetOpenFilename _
                Lib "comdlg32.dll" _
                Alias "GetOpenFileNameA" _
               (pOpenfilename As OPENFILENAME) As Long

Private Declare Function GetSaveFileName _
                Lib "comdlg32.dll" _
                Alias "GetSaveFileNameA" _
                (pOpenfilename As OPENFILENAME) As Long

Dim OFN As OPENFILENAME

Function OpenFile(strVerzeichnis As String, _
                  strTitel As String) As String
  Dim strFilter As String
  Dim strDateinameUndPfad As String
  Dim strDateiname As String
  Dim lngErgebnis As Long

' Angebotene Dateifilter in der Dropdownliste "Dateityp"
  strFilter = "Excel-Arbeitsmappen (*.xls)" & Chr$(0) & _
              "*.XLS" & Chr$(0)

' Vorgegebenes Verzeichnis
  If strVerzeichnis = "" Then
    ' Wenn leer, dann das aktuelle Verzeichnis verwenden
    strVerzeichnis = CurDir$ & Chr$(0)
  Else
    ' "0" an übergebenes Verzeichnis anhängen
    strVerzeichnis = strVerzeichnis & Chr$(0)
  End If

  If strTitel = "" Then
    ' Wenn kein Titel übergeben, Standardtitel festlegen
    strTitel = "Datei-Öffnen:"
  Else
    ' "0" an übergebenen Titel anhängen
    strTitel = strTitel & Chr$(0) 
  End If

' Speicherplatz für Dateinamen & Pfad reservieren
  strDateinameUndPfad = Space$(255) & Chr$(0)

' Speicherplatz für Dateinamen ohne Pfad reservieren
  strDateiname = Space$(255) & Chr$(0)

'Datenstruktur OFN initialisieren
  With OFN
    .lStructSize = Len(OFN)
    .hwndOwner = Application.hWndAccessApp
    .lpstrFilter = strFilter
    .nFilterIndex = 1
    .lpstrFile = strDateinameUndPfad
    .nMaxFile = Len(strDateinameUndPfad)
    .lpstrFileTitle = strDateiname
    .nMaxFileTitle = Len(strDateiname)
    .lpstrInitialDir = strVerzeichnis
    .lpstrTitle = strTitel
    .flags = OFN_FILEMUSTEXIST Or _
             OFN_PATHMUSTEXIST Or _
             OFN_HIDEREADONLY
    .nFileOffset = 0
    .nFileExtension = 0
    .lCustData = 0
    .lpfnHook = 0
    .lpTemplateName = ""
  End With
  lngErgebnis = GetOpenFilename(OFN)

  If lngErgebnis <> 0 Then
      OpenFile = Left(OFN.lpstrFile, _
                      InStr(OFN.lpstrFile, Chr$(0)) - 1)
  Else
      OpenFile = ""
  End If

End Function

Function SaveFile(strVerzeichnis As String, _
                  strTitel As String) As String

  Dim strFilter As String
  Dim strDateinameUndPfad As String
  Dim strDateiname As String
  Dim lngErgebnis As Long

' Angebotene Dateifilter in der Dropdownliste "Dateityp"
  strFilter = "Excel-Arbeitsmappen (*.xls)" & Chr$(0) & _
              "*.XLS" & Chr$(0)

' Vorgegebenes Verzeichnis
  If strVerzeichnis = "" Then
    ' Wenn leer, dann das aktuelle Verzeichnis verwenden
    strVerzeichnis = CurDir$ & Chr$(0)
  Else
    ' "0" an übergebenes Verzeichnis anhängen
    strVerzeichnis = strVerzeichnis & Chr$(0)
  End If

  If strTitel = "" Then
    ' Wenn kein Titel übergeben, Standardtitel festlegen
    strTitel = "Datei speichern:"
  Else
    ' "0" an übergebenen Titel anhängen
    strTitel = strTitel & Chr$(0)
  End If

' Speicherplatz für Dateinamen & Pfad reservieren
  strDateinameUndPfad = Space$(255) & Chr$(0)

' Speicherplatz für Dateinamen ohne Pfad reservieren
  strDateiname = Space$(255) & Chr$(0)

'Datenstruktur OFN initialisieren

  With OFN
    .lStructSize = Len(OFN)
    .hwndOwner = 0&
    .hwndOwner = Application.hWndAccessApp
    .lpstrFilter = strFilter
    .nFilterIndex = 1
    .lpstrFile = strDateinameUndPfad
    .nMaxFile = Len(strDateinameUndPfad)
    .lpstrFileTitle = strDateiname
    .nMaxFileTitle = Len(strDateiname)
    .lpstrInitialDir = strVerzeichnis
    .lpstrTitle = strTitel
    .flags = OFN_HIDEREADONLY Or _
             OFN_OVERWRITEPROMPT
    .nFileOffset = 0
    .nFileExtension = 0
    .lCustData = 0
    .lpfnHook = 0
    .lpTemplateName = ""
  End With

  lngErgebnis = GetSaveFileName(OFN)

  If lngErgebnis <> 0 Then
    SaveFile = Left(OFN.lpstrFile, _
                    InStr(OFN.lpstrFile, Chr$(0)) - 1)
  Else
   SaveFile = ""
  End If

End Function

Die beiden Funktionen "OpenFile()" und "SaveFile()" zeigen jeweils einen Datei-Auswahl-Dialog an und liefern den vom Anwender ausgewählten Dateinamen inklusive Laufwerk und Pfad. Sie können diese Funktionen zum Beispiel einsetzen, um damit per "DoCmd.TransferSpreadsheet" eine Excel-Tabelle in eine Access-Tabelle zu importieren oder um damit eine Access-Tabelle in eine Excel-Arbeitsmappe zu exportieren. Als Parameter übergeben Sie ein Vorgabe-Verzeichnis, aus dem die betreffenden Dateien zum Öffnen angezeigt bzw. in dem eine Datei gespeichert werden soll. Über den zweiten Parameter spezifizieren Sie einen in der Titelleiste des Dialogs anzuzeigenden Hinweis wie zum Beispiel "Tabelle importieren aus Datei:" oder "Tabelle exportieren in Datei:". Das Ergebnis ist entweder eine leere Zeichenkette (""), wenn der Anwender auf Abbrechen geklickt hat oder der gewählte Dateiname inklusive Laufwerk und Pfad.

Wenn Sie andere Dateitypen als "XLS" zur Auswahl anzeigen möchten, passen Sie die Funktionen in der folgenden Zeile, die jeweils identisch ist, an:

' Angebotene Dateifilter in der Dropdownliste "Dateityp"
  strFilter = "Excel-Arbeitsmappen (*.xls)" & _
               Chr$(0) & "*.XLS" & Chr$(0)

Die Angabe besteht aus zwei Teilen, die jeweils durch ein Nullbyte "Chr$(0)" zu trennen sind. Der vordere Teil definiert den Text, der in der Liste "Dateityp" angezeigt wird, der hintere Teil definiert die Dateimaske für die Auswahl, wobei Wildcards (*, ?) erlaubt sind. Sie können hier auch mehrere Dateimasken definieren, indem Sie sie durch Semikolon trennen:

strFilter = "Word-Dateien (*.doc;*.dot)" & _
            Chr$(0) & "*.DOC;*.DOT" & Chr$(0)

Mit Office XP hat Microsoft wie bereits erwähnt das Objekt "FileDialogs" eingeführt, das ohne externe API-Funktionen und komplexe Datenstrukturen auskommt und zudem komfortabler zu handhaben ist. Die oben gezeigten Funktionen können jedoch auch problemlos mit Access 2003 und 2002/XP eingesetzt werden! Das FileDialog-Objekt kann generell in allen Office-Anwendungen eingesetzt werden. Es gehört jedoch nicht zum jeweiligen Objektmodell der Office-Anwendung, sondern ist zentral über die Objekt-Bibliothek "Microsoft Office X.x Object Library" verfügbar (X.x= 11.0 für Office 2003 und 10.0 für Office XP). Bevor Sie in Ihren Lösungen mit diesem Objekt arbeiten können, ist also zunächst über das Menü Extras-Verweise eine Referenz auf diese Bibliothek zu setzen.

Die auf "FileDialogs" basierenden Funktionen "OpenFile()" und "SaveFile()" sind wie folgt aufgebaut:

Function OpenFile(strTitle As String, _
                  strPath As String, _
                  strButton As String) As String
  Dim fDlg As FileDialog

  Set fDlg = Application.FileDialog(msoFileDialogOpen)
  With fDlg
    .Title = strTitle
    .InitialFileName = strPath
    .ButtonName = strButton
    .Show
    If .SelectedItems.Count = 0 Then Exit Function
    OpenFile = .SelectedItems(1)
  End With

End Function

Function SaveFile(strTitle As String, _
                  strPath As String, _
                  strButton As String) As String
  Dim fDlg As FileDialog

  Set fDlg = Application.FileDialog(msoFileDialogSaveAs)
  With fDlg
    .Title = strTitle
    .InitialFileName = strPath
    .ButtonName = strButton
    .Show
    If .SelectedItems.Count = 0 Then Exit Function
    SaveFile = .SelectedItems(1)
  End With

End Function

Als Parameter übergeben Sie neben dem Startverzeichnis für die Auswahl und dem Text für die Anzeige in der Titelleiste einen Text für die Beschriftung der Auswahl-Schaltfläche. Statt Öffnen" können Sie zum Beispiel "Importieren" oder "Exportieren" anzeigen lassen. Das Ergebnis beider Funktionen ist eine leere Zeichenkette "", wenn auf Abbrechen geklickt wurde oder der Name der gewählten Datei inklusive Laufwerk und Pfad.

